

A Symbolic Solution of a 3D Affine Transformation

Béla Paláncz
Zaletnyik Piroska

We demonstrate a symbolic elimination technique to solve a nine-parameter 3D affine transformation when only three known points in both systems are given. The system of nine equations is reduced to six by subtracting the equations and eliminating the translation parameters. From these six equations, five variables are eliminated using a Gröbner basis to get a quadratic univariate polynomial, from which the solution can be expressed symbolically. The main advantage of this result is that we do not need to guess initial values of the nine parameters, which is necessary in the case of the traditional solution of the nonlinear system of equations. This result can be useful in geodesy, robotics, and photogrammetry when occasionally only three known points in both systems are given or when a Gauss–Jacobi combinatorial solution may be required for certain reasons, for example detecting outliers by using variance-covariance matrices.

■ Introduction

Three-dimensional data transformations play a central role in contemporary Euclidean point positioning. In precise positioning with a global positioning system (GPS), coordinates given in the World Geodetic System 1984 (WGS84) often have to be transformed into a local geodetic coordinate system. The transformation between the traditional terrestrial coordinate system and the derived network of satellite observations is a difficult task due to the heterogeneity of the data.

Due to the distortions between the traditional terrestrial and the GPS-derived networks, the seven-parameter similarity transformations may not offer satisfactory precision in some cases. To reduce the remaining residuals after the transformation, other transformation models with more parameters can be used. The nine-parameter affine transformation is not only a logical extension but even a generalization of the seven-parameter similarity transformation model; when the three scale parameters are equal, we get the similarity

transformation model. This transformation modifies the Helmert $C_7(3)$ transformation, where three different scales are used in the corresponding coordinate axes instead of only one scale factor. Späth [1] estimated the model parameters by using a numerical minimization technique of the residual vector and Papp [2] used a linearized least-squares method. Watson [3] pointed out that the Gauss–Newton method or its variants can be easily implemented for the nine-parameter problem using separation of variables and iteration with respect to the rotation parameters alone, while other parameters can be calculated via a simple linear least-squares solution. The method he suggested is analogous to other methods for separated least-squares problems, which go back at least to Golub and Pereyra [4]. The nine-parameter affine transformation is also included in some coordinate-transformation software developed at the request of GPS users (e.g., see Mathes [5] and Fröhlich and Bröker [6]). Here we should mention transformation models with more than nine parameters. Wolfrum [7] even added one more parameter to the previous nine, a (horizontal) direction of maximal scale distortion. Grafarend and Kampmann [8] applied a ten-parameter conformal group for geodetic data transformation employing maximum likelihood estimations for numerical estimation of the parameter values. There are other models with even more parameters, like polynomial transformations (Volgyesi et al [9], Cai and Grafarend [10]) and models using artificial neural networks (Barsi [11], Zaletnyik [12]).

■ Definition of the Problem

A 3D affine transformation is one possible generalization of the $C_7(3, 3)$ Helmert transformation, using three different scale parameters s_1, s_2, s_3 instead of a single one. In this case the scale factors can be modeled by a diagonal matrix W ,

$$\begin{pmatrix} x_i \\ y_i \\ z_i \end{pmatrix} = W R \begin{pmatrix} X_i \\ Y_i \\ Z_i \end{pmatrix} + \begin{pmatrix} X_0 \\ Y_0 \\ Z_0 \end{pmatrix}, \quad (1)$$

where

$$W = \begin{pmatrix} s_1 & 0 & 0 \\ 0 & s_2 & 0 \\ 0 & 0 & s_3 \end{pmatrix}, \quad (2)$$

X_0, Y_0, Z_0 are the three translation parameters, and R is the rotation matrix expressed in terms of the skew-symmetric matrix S (see Awange and Grafarend [13]),

$$S = \begin{pmatrix} 0 & -c & -b \\ c & 0 & -a \\ -b & a & 0 \end{pmatrix}. \quad (3)$$

The rotation matrix is

$$R = (I_3 - S)^{-1} (I_3 + S), \quad (4)$$

where I_3 is the 3×3 identity matrix.

Let $\sigma_i = \frac{1}{s_i}$ and $\Omega = W^{-1}$.

To determine the nine parameters of the transformation ($a, b, c, X_0, Y_0, Z_0, s_1, s_2, s_3$) we need three noncollinear points with known coordinates in both coordinate systems. Instead of the scale parameters s_1, s_2, s_3 , we use $\sigma_1, \sigma_2, \sigma_3$ to get simpler equations.

The following commands define the matrices in *Mathematica*.

$$\mathbf{S} = \begin{pmatrix} \mathbf{0} & -\mathbf{c} & \mathbf{b} \\ \mathbf{c} & \mathbf{0} & -\mathbf{a} \\ -\mathbf{b} & \mathbf{a} & \mathbf{0} \end{pmatrix};$$

$$\mathbf{I}_3 = \text{IdentityMatrix}[3];$$

$$\mathbf{R} = \text{Inverse}[(\mathbf{I}_3 - \mathbf{S})] \cdot (\mathbf{I}_3 + \mathbf{S}) // \text{Simplify};$$

$$\text{MatrixForm}[\mathbf{R}]$$

$$\begin{pmatrix} \frac{1+a^2-b^2-c^2}{1+a^2+b^2+c^2} & \frac{2ab-2c}{1+a^2+b^2+c^2} & \frac{2(b+ac)}{1+a^2+b^2+c^2} \\ \frac{2(ab+c)}{1+a^2+b^2+c^2} & \frac{1-a^2+b^2-c^2}{1+a^2+b^2+c^2} & -\frac{2(a-bc)}{1+a^2+b^2+c^2} \\ \frac{2(-b+ac)}{1+a^2+b^2+c^2} & \frac{2(a+bc)}{1+a^2+b^2+c^2} & \frac{1-a^2-b^2+c^2}{1+a^2+b^2+c^2} \end{pmatrix}$$

$$\Omega = \begin{pmatrix} \sigma_1 & \mathbf{0} & \mathbf{0} \\ \mathbf{0} & \sigma_2 & \mathbf{0} \\ \mathbf{0} & \mathbf{0} & \sigma_3 \end{pmatrix};$$

Expressing the rotation matrix with the skew-symmetric matrix and using the inverse of the scale matrix Ω , this is the nonlinear system.

$$\text{Table} \left[\begin{pmatrix} \mathbf{f}_{1+3} (i-1) \\ \mathbf{f}_{2+3} (i-1) \\ \mathbf{f}_{3+3} (i-1) \end{pmatrix} = \left((\mathbf{I}_3 - \mathbf{S}) \cdot \Omega \cdot \begin{pmatrix} \mathbf{x}_i \\ \mathbf{y}_i \\ \mathbf{z}_i \end{pmatrix} - (\mathbf{I}_3 + \mathbf{S}) \cdot \begin{pmatrix} \mathbf{x}_i \\ \mathbf{y}_i \\ \mathbf{z}_i \end{pmatrix} - (\mathbf{I}_3 - \mathbf{S}) \cdot \Omega \cdot \begin{pmatrix} \mathbf{x}_0 \\ \mathbf{y}_0 \\ \mathbf{z}_0 \end{pmatrix} \right), \{i, 0, 3\} \right];$$

```

eq = Table[{"f" i, "=", fi}, {i, 1, 9}]; TableForm[eq]

f1 = -X1 + c Y1 - b Z1 + x1 σ1 - X0 σ1 + c Y1 σ2 - c Y0 σ2 - b z1 σ3 + b Z0 σ3
f2 = -c X1 - Y1 + a Z1 - c x1 σ1 + c X0 σ1 + y1 σ2 - Y0 σ2 + a z1 σ3 - a Z0 σ3
f3 = b X1 - a Y1 - Z1 + b x1 σ1 - b X0 σ1 - a y1 σ2 + a Y0 σ2 + z1 σ3 - Z0 σ3
f4 = -X2 + c Y2 - b Z2 + x2 σ1 - X0 σ1 + c y2 σ2 - c Y0 σ2 - b z2 σ3 + b Z0 σ3
f5 = -c X2 - Y2 + a Z2 - c x2 σ1 + c X0 σ1 + y2 σ2 - Y0 σ2 + a z2 σ3 - a Z0 σ3
f6 = b X2 - a Y2 - Z2 + b x2 σ1 - b X0 σ1 - a y2 σ2 + a Y0 σ2 + z2 σ3 - Z0 σ3
f7 = -X3 + c Y3 - b Z3 + x3 σ1 - X0 σ1 + c y3 σ2 - c Y0 σ2 - b z3 σ3 + b Z0 σ3
f8 = -c X3 - Y3 + a Z3 - c x3 σ1 + c X0 σ1 + y3 σ2 - Y0 σ2 + a z3 σ3 - a Z0 σ3
f9 = b X3 - a Y3 - Z3 + b x3 σ1 - b X0 σ1 - a y3 σ2 + a Y0 σ2 + z3 σ3 - Z0 σ3

```

■ Elimination of the Translation Parameters

Solutions of the nonlinear system of equations in geodesy have received wide coverage, as evidenced in the works of Bancroft [14], Grafarend and Schaffrin [15], Singer et al. [16], Kleusberg [17, 18], Lichtenegger [19], and Awange [20]. Similarly to the $C_7(3)$ problem (see Awange and Grafarend [21]), we can reduce the system by subtracting the equations, eliminating the translation parameters.

```

sysR = {f1 - f7, f4 - f7, f2 - f8, f5 - f8, f3 - f9, f6 - f9} //
FullSimplify

{-X1 + X3 + (x1 - x3) σ1 + c (Y1 - Y3 + (y1 - y3) σ2) +
 b (-Z1 + Z3 + (-z1 + z3) σ3), -X2 + X3 + (x2 - x3) σ1 +
 c (Y2 - Y3 + (y2 - y3) σ2) + b (-Z2 + Z3 + (-z2 + z3) σ3),
 -Y1 + Y3 + c (-X1 + X3 + (-x1 + x3) σ1) + (Y1 - Y3) σ2 +
 a (Z1 - Z3 + (z1 - z3) σ3), -Y2 + Y3 + c (-X2 + X3 + (-x2 + x3) σ1) +
 (Y2 - Y3) σ2 + a (Z2 - Z3 + (z2 - z3) σ3),
 -Z1 + Z3 + b (X1 - X3 + (x1 - x3) σ1) + a (-Y1 + Y3 + (-y1 + y3) σ2) +
 (z1 - z3) σ3, -Z2 + Z3 + b (X2 - X3 + (x2 - x3) σ1) +
 a (-Y2 + Y3 + (-y2 + y3) σ2) + (z2 - z3) σ3}

```

■ Introduction of Relative Coordinates

Employing a Gröbner basis, the reduced equation system can easily be reduced further if we introduce some new variables. Let us call these relative coordinates.

It is clear that introducing the following variables can simplify the description of the system.

```
newVarsA = {x12 → x1 - x2, x13 → x1 - x3, x23 → x2 - x3,
  y12 → y1 - y2, y13 → y1 - y3, y23 → y2 - y3,
  z12 → z1 - z2, z13 → z1 - z3, z23 → z2 - z3,
  X12 → X1 - X2, X13 → X1 - X3, X23 → X2 - X3,
  Y12 → Y1 - Y2, Y13 → Y1 - Y3, Y23 → Y2 - Y3,
  Z12 → Z1 - Z2, Z13 → Z1 - Z3, Z23 → Z2 - Z3};
```

This is the new system.

```
g1 = -X13 + c Y13 - b Z13 + x13 σ1 + c y13 σ2 - b z13 σ3;
g2 = -X23 + c Y23 - b Z23 + x23 σ1 + c y23 σ2 - b z23 σ3;
g3 = -c X13 - Y13 + a Z13 - c x13 σ1 + y13 σ2 + a z13 σ3;
g4 = -c X23 - Y23 + a Z23 - c x23 σ1 + y23 σ2 + a z23 σ3;
g5 = b X13 - a Y13 - Z13 + b x13 σ1 - a y13 σ2 + z13 σ3;
g6 = b X23 - a Y23 - Z23 + b x23 σ1 - a y23 σ2 + z23 σ3;
```

We verify the equivalence.

```
{g1, g2, g3, g4, g5, g6} /. newVarsA) - sysR // FullSimplify
{0, 0, 0, 0, 0, 0}
```

■ Symbolic Solution for the Scale Parameters ($\sigma_1, \sigma_2, \sigma_3$)

Our intention is to reduce the computation time of the three-point solution because (eg., in the case of a Gauss–Jacobi method for $N > 3$ points) it is very important to use an effective method for solving one of the many combinations. This reduction can be done by a reduced Gröbner basis.

```
gbσ1 = GroebnerBasis[{g1, g2, g3, g4, g5, g6}, σ1,
  {a, b, c, σ2, σ3}, MonomialOrder → EliminationOrder]
{-X232 y13 z13 + X13 X23 y23 z13 + Y13 y23 Y23 z13 -
  y13 Y232 z13 + X13 X23 y13 z23 - X132 y23 z23 - Y132 y23 z23 +
  y13 Y13 Y23 z23 - y23 Z132 z23 + y23 z13 Z13 Z23 +
  y13 Z13 z23 Z23 - y13 z13 Z232 + x232 y13 z13 σ12 -
  x13 x23 y23 z13 σ12 - x13 x23 y13 z23 σ12 + x132 y23 z23 σ12}
```

```
gb $\sigma_2$  = GroebnerBasis[{g1, g2, g3, g4, g5, g6},  $\sigma_2$ ,
  {a, b, c,  $\sigma_1$ ,  $\sigma_3$ }, MonomialOrder  $\rightarrow$  EliminationOrder]
```

$$\{-X_{13} x_{23} X_{23} z_{13} + x_{13} X_{23}^2 z_{13} - x_{23} Y_{13} Y_{23} z_{13} +$$

$$x_{13} Y_{23}^2 z_{13} + X_{13}^2 x_{23} z_{23} - x_{13} X_{13} X_{23} z_{23} + x_{23} Y_{13}^2 z_{23} -$$

$$x_{13} Y_{13} Y_{23} z_{23} + x_{23} Z_{13}^2 z_{23} - x_{23} z_{13} Z_{13} Z_{23} -$$

$$x_{13} Z_{13} z_{23} Z_{23} + x_{13} z_{13} Z_{23}^2 + x_{23} y_{13} y_{23} z_{13} \sigma_2^2 -$$

$$x_{13} y_{23}^2 z_{13} \sigma_2^2 - x_{23} y_{13}^2 z_{23} \sigma_2^2 + x_{13} y_{13} y_{23} z_{23} \sigma_2^2\}$$

```
gb $\sigma_3$  = GroebnerBasis[{g1, g2, g3, g4, g5, g6},  $\sigma_3$ ,
  {a, b, c,  $\sigma_1$ ,  $\sigma_2$ }, MonomialOrder  $\rightarrow$  EliminationOrder]
```

$$\{X_{13} x_{23} X_{23} y_{13} - x_{13} X_{23}^2 y_{13} - X_{13}^2 x_{23} y_{23} +$$

$$x_{13} X_{13} X_{23} y_{23} - x_{23} Y_{13}^2 y_{23} + x_{23} y_{13} Y_{13} Y_{23} +$$

$$x_{13} Y_{13} y_{23} Y_{23} - x_{13} y_{13} Y_{23}^2 - x_{23} y_{23} Z_{13}^2 + x_{23} y_{13} Z_{13} Z_{23} +$$

$$x_{13} y_{23} Z_{13} Z_{23} - x_{13} y_{13} Z_{23}^2 + x_{23} y_{23} z_{13} \sigma_3^2 -$$

$$x_{23} y_{13} z_{13} z_{23} \sigma_3^2 - x_{13} y_{23} z_{13} z_{23} \sigma_3^2 + x_{13} y_{13} z_{23}^2 \sigma_3^2\}$$

The polynomials providing the solution for $\sigma_1, \sigma_2, \sigma_3$ are quadratic, therefore their solution can be expressed in analytic form (in all of these three cases, only the positive root is correct).

```
sol $\sigma_1$  = (Solve[gb $\sigma_1$  == 0,  $\sigma_1$ ] [[2]] // Simplify)
```

$$\left\{ \sigma_1 \rightarrow \left(\sqrt{ \left(X_{23}^2 y_{13} z_{13} + y_{13} Y_{23}^2 z_{13} + \right.} \right. \right.$$

$$\left. \left. X_{13}^2 y_{23} z_{23} + Y_{13}^2 y_{23} z_{23} + y_{23} Z_{13}^2 z_{23} - \right. \right.$$

$$\left. \left. X_{13} X_{23} (y_{23} z_{13} + y_{13} z_{23}) - Y_{13} Y_{23} (y_{23} z_{13} + y_{13} z_{23}) - \right. \right.$$

$$\left. \left. y_{23} z_{13} Z_{13} Z_{23} - y_{13} Z_{13} z_{23} Z_{23} + y_{13} z_{13} Z_{23}^2 \right) \right) /$$

$$\left(\sqrt{ (x_{23} y_{13} - x_{13} y_{23}) (x_{23} z_{13} - x_{13} z_{23}) } \right) \left. \right\}$$

```
sol $\sigma_2$  = (Solve[gb $\sigma_2$  == 0,  $\sigma_2$ ] [[2]] // Simplify)
```

$$\left\{ \sigma_2 \rightarrow \left(\sqrt{ \left(-X_{13}^2 x_{23} z_{23} + X_{13} X_{23} (x_{23} z_{13} + x_{13} z_{23}) + x_{23} (Y_{13} Y_{23} z_{13} - \right. \right.} \right.$$

$$\left. \left. Y_{13}^2 z_{23} - Z_{13}^2 z_{23} + z_{13} Z_{13} Z_{23}) - x_{13} (X_{23}^2 z_{13} + \right. \right.$$

$$\left. \left. Y_{23}^2 z_{13} - Y_{13} Y_{23} z_{23} - Z_{13} z_{23} Z_{23} + z_{13} Z_{23}^2 \right) \right) \right) /$$

$$\left(\sqrt{ - (x_{23} y_{13} - x_{13} y_{23}) (-y_{23} z_{13} + y_{13} z_{23}) } \right) \left. \right\}$$

```
solo3 = (Solve[gbσ3 == 0, σ3][[2]] // Simplify)
```

$$\left\{ \sigma_3 \rightarrow \frac{\left(\sqrt{\left(x_{13}^2 x_{23} y_{23} - x_{13} x_{23} (x_{23} y_{13} + x_{13} y_{23}) + x_{23} (y_{13}^2 y_{23} - y_{13} y_{13} y_{23} + y_{23} z_{13}^2 - y_{13} z_{13} z_{23}) + x_{13} (x_{23}^2 y_{13} - y_{13} y_{23} y_{23} + y_{13} y_{23}^2 - y_{23} z_{13} z_{23} + y_{13} z_{23}^2) \right)} \right)}{\left(\sqrt{(x_{23} z_{13} - x_{13} z_{23}) (y_{23} z_{13} - y_{13} z_{23})} \right)} \right\}$$

The univariate polynomial for $\sigma_1, \sigma_2, \sigma_3$ can also be computed by employing the accelerated Dixon resultant by the early discovery factors algorithm (Dixon's EDF), which was suggested and implemented in the computer algebra system Fermat by Lewis [22].

■ Symbolic Solution for the Rotation Parameters (a, b, c)

The parameters of the skew-symmetric matrix can also be computed using a reduced Gröbner basis. Let us express **a** from equations g_5 and g_6 .

```
{grba} = GroebnerBasis[{g5, g6}, {a, b}, {b}]
```

$$\{-a x_{23} y_{13} + a x_{13} y_{23} - x_{23} z_{13} + x_{13} z_{23} - a x_{23} y_{13} \sigma_1 + a x_{13} y_{23} \sigma_1 - x_{23} z_{13} \sigma_1 + x_{13} z_{23} \sigma_1 - a x_{23} y_{13} \sigma_2 + a x_{13} y_{23} \sigma_2 - a x_{23} y_{13} \sigma_1 \sigma_2 + a x_{13} y_{23} \sigma_1 \sigma_2 + x_{23} z_{13} \sigma_3 - x_{13} z_{23} \sigma_3 + x_{23} z_{13} \sigma_1 \sigma_3 - x_{13} z_{23} \sigma_1 \sigma_3\}$$

Then this finds **a**.

```
sola = Solve[grba == 0, a] // Simplify // Flatten
```

$$\{a \rightarrow \frac{(x_{23} z_{13} - x_{13} z_{23} + (-x_{23} z_{13} + x_{13} z_{23}) \sigma_3 + \sigma_1 (x_{23} z_{13} - x_{13} z_{23} + (-x_{23} z_{13} + x_{13} z_{23}) \sigma_3)) / (-x_{23} y_{13} + x_{13} y_{23} + (-x_{23} y_{13} + x_{13} y_{23}) \sigma_2 + \sigma_1 (-x_{23} y_{13} + x_{13} y_{23} + (-x_{23} y_{13} + x_{13} y_{23}) \sigma_2))}{x_{13} + x_{13} \sigma_1}\}$$

The parameter **b** is given by one of the two equations, let us say from g_5 .

```
solb = Solve[g5 == 0, b] // Simplify // Flatten
```

$$\{b \rightarrow \frac{a y_{13} + z_{13} + a y_{13} \sigma_2 - z_{13} \sigma_3}{x_{13} + x_{13} \sigma_1}\}$$

The parameter c comes from g_1 .

```
solc = Solve[g1 == 0, c] // Simplify // Flatten
```

$$\left\{ c \rightarrow \frac{X_{13} + b Z_{13} - x_{13} \sigma_1 + b z_{13} \sigma_3}{Y_{13} + y_{13} \sigma_2} \right\}$$

■ Symbolic Solution for the Translation Parameters (x_0, y_0, z_0)

The translation parameters can be similarly computed, but now from the original system of equations, f_i .

```
{grbX0} = GroebnerBasis[{f1, f2, f3}, {X0, Y0, Z0}, {Y0, Z0}]
```

$$\left\{ X_1 + a^2 X_1 - b^2 X_1 - c^2 X_1 + 2 a b Y_1 - 2 c Y_1 + 2 b Z_1 + 2 a c Z_1 - x_1 \sigma_1 - a^2 x_1 \sigma_1 - b^2 x_1 \sigma_1 - c^2 x_1 \sigma_1 + X_0 \sigma_1 + a^2 X_0 \sigma_1 + b^2 X_0 \sigma_1 + c^2 X_0 \sigma_1 \right\}$$

```
solX0 = Solve[grbX0 == 0, X0] // Simplify // Flatten
```

$$\left\{ X_0 \rightarrow \frac{1}{(1 + a^2 + b^2 + c^2) \sigma_1} \left((-1 - a^2 + b^2 + c^2) X_1 + (-2 a b + 2 c) Y_1 - 2 b Z_1 - 2 a c Z_1 + x_1 \sigma_1 + a^2 x_1 \sigma_1 + b^2 x_1 \sigma_1 + c^2 x_1 \sigma_1 \right) \right\}$$

```
{grbY0} = GroebnerBasis[{f1, f2, f3}, {X0, Y0, Z0}, {X0, Z0}]
```

$$\left\{ 2 a b X_1 + 2 c X_1 + Y_1 - a^2 Y_1 + b^2 Y_1 - c^2 Y_1 - 2 a Z_1 + 2 b c Z_1 - y_1 \sigma_2 - a^2 y_1 \sigma_2 - b^2 y_1 \sigma_2 - c^2 y_1 \sigma_2 + Y_0 \sigma_2 + a^2 Y_0 \sigma_2 + b^2 Y_0 \sigma_2 + c^2 Y_0 \sigma_2 \right\}$$

```
solY0 = Solve[grbY0 == 0, Y0] // Simplify // Flatten
```

$$\left\{ Y_0 \rightarrow \frac{1}{(1 + a^2 + b^2 + c^2) \sigma_2} \left(-2 (a b + c) X_1 + (-1 + a^2 - b^2 + c^2) Y_1 + 2 a Z_1 - 2 b c Z_1 + y_1 \sigma_2 + a^2 y_1 \sigma_2 + b^2 y_1 \sigma_2 + c^2 y_1 \sigma_2 \right) \right\}$$


```
solz0 = Solve[f1 == 0, z0] // Simplify // Flatten
```

$$\left\{ z_0 \rightarrow \frac{X_1 - c Y_1 + b Z_1 - x_1 \sigma_1 + X_0 \sigma_1 - c Y_0 \sigma_2 + c Y_0 \sigma_2 + b Z_1 \sigma_3}{b \sigma_3} \right\}$$

■ Numerical Example

Let us consider the numerical values of three Hungarian points in the system of ETRS89 (European Terrestrial Reference System 1989) $(x_1, y_1, z_1, x_1, y_1, z_1)$ and in the local Hungarian system HD72 (Hungarian Data 1972) $(X_1, Y_1, Z_1, X_1, Y_1, Z_1, X_1, Y_1, Z_1)$.

```
numericalValues = {
  x1 -> 4 171 409.677, x2 -> 4 146 957.889, x3 -> 3 955 632.880,
  y1 -> 1 470 823.777, y2 -> 1 277 033.850, y3 -> 1 611 863.197,
  z1 -> 4 580 140.907, z2 -> 4 659 439.264, z3 -> 4 720 991.316,
  X1 -> 4 171 352.311, X2 -> 4 146 901.301, X3 -> 3 955 575.649,
  Y1 -> 1 470 893.887, Y2 -> 1 277 104.509, Y3 -> 1 611 933.124,
  Z1 -> 4 580 150.178, Z2 -> 4 659 448.287, Z3 -> 4 721 000.952
};
```

□ Using the Analytical Form of the Symbolic Solution

Now, we proceed with the numerical computation of the parameters, substituting the numerical values into the symbolic formulas.

```
{tσ1, σ1} =
  SetPrecision[(σ1 /. solσ1 /. newVarsA /. numericalValues),
    16] // Chop // Timing
{0.00051, 1.0000054081636032}

{tσ2, σ2} =
  SetPrecision[(σ2 /. solσ2 /. newVarsA /. numericalValues),
    16] // Chop // Timing
{0.000452, 0.9999978437466494}

{tσ3, σ3} =
  SetPrecision[(σ3 /. solσ3 /. newVarsA /. numericalValues),
    16] // Chop // Timing
{0.00043, 0.9999892916567600}
```

Here are the numerical values of the scale parameters.

```
sσ = {σ1 -> σ1, σ2 -> σ2, σ3 -> σ3}
{σ1 → 1.0000054081636032,
 σ2 → 0.9999978437466494, σ3 → 0.9999892916567600}
```

We continue similarly.

```
{ta, aa} =
SetPrecision[(a /. sola /. newVarsA /. sσ /. numericalValues),
16] // Chop // Timing
{0.000463, 9.269080332103913 × 10-7}

{tb, bb} =
SetPrecision[
(b /. solb /. newVarsA /. sσ /. {a -> aa} /.
numericalValues), 16] // Chop // Timing
{0.0002, -4.946616458808338 × 10-6}

{tc, cc} =
SetPrecision[
(c /. solc /. newVarsA /. sσ /. {b -> bb} /.
numericalValues), 16] // Chop // Timing
{0.000289, -3.243922351302771 × 10-7}
```

These are the parameters in the skew-symmetric matrix.

```
sabc = {a -> aa, b -> bb, c -> cc}
{a → 9.269080332103913 × 10-7,
 b → -4.946616458808338 × 10-6, c → -3.243922351302771 × 10-7}
```

These are the translation parameters.

```
{tX0, X0} =
SetPrecision[
(X0 /. solX0 /. newVarsA /. sσ /. sabc /. numericalValues),
16] // Chop // Timing
{0.000454, 124.2834144979486}
```

```

{tY0, Y0} =
  SetPrecision[
    (Y0 /. solY0 /. newVarsA /. sσ /. sabc /. numericalValues),
    16] // Chop // Timing
{0.000298, -62.08451159187030}

{tZ0, Z0} =
  SetPrecision[
    (Z0 /. solZ0 /. newVarsA /. sσ /. sabc /.
     {X0 -> X0, Y0 -> Y0} /. numericalValues), 16] // Chop //
  Timing
{0.000262, -102.3123880882032}

```

The total time for the numerical computation of the parameters from the symbolic solution is practically zero.

```

Total[{tσ1, tσ2, tσ3, ta, tb, tc, tX0, tY0, tZ0}]
0.003358

```

□ Using Numerical Global Polynomial Solver

In order to compare this result with the direct numerical solution, we solve the original system with NSolve.

```

Neqs = {f1, f2, f3, f4, f5, f6, f7, f8, f9} /. numericalValues;

```

The solution provides all roots.

```

sol = NSolve[Neqs, {a, b, c, σ1, σ2, σ3, X0, Y0, Z0}] // Timing
{0.13485,
 { {a → -1.21157 × 10-6, b → 1.17632, c → -0.35517, σ1 → -1.00001,
   σ2 → 0.999998, σ3 → 0.999989, X0 → 8.0335 × 106,
   Y0 → 2.85318 × 106, Z0 → 9.44997 × 106},
 {a → -3.31199, b → -3.41124 × 10-6, c → 2.81556,
   σ1 → 1.00001, σ2 → -0.999998, σ3 → 0.999989,
   X0 → 8.0335 × 106, Y0 → 2.85318 × 106, Z0 → 9.44997 × 106},
 {a → 0.301933, b → -0.850109, c → -1.02997 × 10-6,
   σ1 → 1.00001, σ2 → 0.999998, σ3 → -0.999989,
   X0 → 8.0335 × 106, Y0 → 2.85318 × 106, Z0 → 9.44997 × 106},
 {a → 825 376., b → 293 148., c → 970 906., σ1 → -1.00001,

```

```

σ2 → -0.999998, σ3 → -0.999989, X0 → 8.0335 × 106,
Y0 → 2.85318 × 106, Z0 → 9.44997 × 106},
{a → 15.2489, b → 2.85737, c → 3.08269 × 106,
σ1 → -1.00001, σ2 → -0.999998, σ3 → 0.999989,
X0 → 124.283, Y0 → -62.0845, Z0 → -102.313},
{a → -1.07886 × 106, b → -0.349972, c → 5.33669,
σ1 → 1.00001, σ2 → -0.999998, σ3 → -0.999989,
X0 → 124.283, Y0 → -62.0845, Z0 → -102.313},
{a → -0.0655786, b → 202.158., c → -0.187382,
σ1 → -1.00001, σ2 → 0.999998, σ3 → -0.999989,
X0 → 124.283, Y0 → -62.0845, Z0 → -102.313},
{a → 9.26908 × 10-7, b → -4.94662 × 10-6, c → -3.24392 × 10-7,
σ1 → 1.00001, σ2 → 0.999998, σ3 → 0.999989,
X0 → 124.283, Y0 → -62.0845, Z0 → -102.313}}}]

```

From the eight solutions we need only the one where the scale variables are positive, $\sigma_i > 0$.

```

sol3NSolve =
Select[sol[[2]],
(#[[4, 2]] > 0) & (#[[5, 2]] > 0) & (#[[6, 2]] > 0) &] // Timing
{0.000125,
{{a → 9.26908 × 10-7, b → -4.94662 × 10-6, c → -3.24392 × 10-7,
σ1 → 1.00001, σ2 → 0.999998, σ3 → 0.999989,
X0 → 124.283, Y0 → -62.0845, Z0 → -102.313}}}]

```

The results of the numerical and symbolic solutions are the same, but the numerical solution is slower than the symbolic solution (which is practically zero instantaneous) and the numerical solution gives more than one solution, from which the single good solution must be selected.

■ Conclusion

A fully symbolic solution was suggested to solve the 3D affine data transformation problem. We demonstrated that an elimination technique (like the reduced Gröbner basis) can be used effectively to solve this problem. The symbolic reduction can be carried out “offline,” only once, so it does not influence the computing time. This feature can be very useful when one needs to solve a three-point problem multiple times, for example in the case of the Gauss–Jacobi solution of an N -point problem. In addition, the result of the three-point problem can provide a good initial guess for solving an N -point problem with local numerical methods.

Numerical computations were carried out with *Mathematica 7* running on an HP xw 4100 workstation with Windows XP operating system, 3 GHz P4 Intel processor, and 1 GB RAM.

■ Acknowledgments

The authors are grateful to the reviewer for his/her remarks and suggestions and are indebted also to Daniel Lichtblau (Wolfram Research) for his valuable help in calling their attention to the proper application of the reduced Gröbner basis.

■ References

- [1] H. Späth, “A Numerical Method for Determining the Spatial Helmert Transformation in the Case of Different Scale Factors,” *Zeitschrift für Geodäsie, Geoinformation und Landmanagement*, **129**, 2004 pp. 225–257.
- [2] E. Papp and L. Szucs, “Transformation Methods of the Traditional and Satellite Based Networks,” (in Hungarian with English abstract), *Geomatikai Közlemények*, **VIII**, 2005 pp. 85–92.
- [3] G. A. Watson, “Computing Helmert Transformations,” *Journal of Computational and Applied Mathematics*, **197**(2), 2006 pp. 387–394. dx.doi.org/doi:10.1016/j.cam.2005.06.04.
- [4] G. H. Golub and V. Pereyra, “The Differentiation of Pseudo-Inverses and Nonlinear Least Squares Problems Whose Variables Separate,” *SIAM Journal on Numerical Analysis*, **10**(2), 1973 pp. 413–432. www.jstor.org/stable/2156365.
- [5] A. Mathes, *EasyTrans Pro-Edition: Professionelle Koordinatentransformation für Navigation, Vermessung und GIS* (CD-ROM mit Benutzerhandbuch), Berlin: Herbert Wichmann Verlag, 2002.
- [6] H. Fröhlich and G. Bröker. “Trafox Version 2.1: 3d-kartesische Helmert-Transformation.” (2003).
- [7] O. Wolfrum, “Merging Terrestrial and Satellite Networks by a Ten-Parameter Transformation Model,” *Manuscripta Geodaetica*, **17**(4), 1992 pp. 210–214.
- [8] E. Grafarend and G. Kampmann, “ $C_{10}(3)$: The Ten-Parameter Conformal Group as a Datum Transformation in Three-Dimensional Euclidean Space,” *Zeitschrift für Vermessungswesen*, **2**, 1996 pp. 68–77. www.uni-stuttgart.de/gi/research/paper/1996/Grafarend_Kampmann.pdf.

- [9] L. Völgyesi, G. Tóth, and J. Varga, “Conversions between Hungarian Map Projection Systems,” *Periodica Polytechnica Civil Engineering*, **40**(1), 1996 pp. 73–83. www.pp.bme.hu/ci.
- [10] J. Cai and E. W. Grafarend, “Systematische Analyse der Transformation zwischen Gauss–Krüger” (poster presentation), Koordinaten/DHDN und UTM—Koordinaten/ETRS89 angewandt auf Baden-Württemberg, Geodätische Woche 2004, Stuttgart, 2004.
- [11] A. Barsi, “Performing Coordinate Transformation by Artificial Neural Network,” *Allgemeine Vermessungs-Nachrichten*, **108**(4), 2001 pp. 134–137. www.wichmann-verlag.de/avn-ausgabe-04-2001/performing-coordinate-transformation-by-artificial-neural-network.html.
- [12] P. Zaletnyik, “Coordinate Transformation with Neural Networks and with Polynomials in Hungary,” in *Proceedings of International Symposium on Modern Technologies, Education and Professional Practice in Geodesy and Related Fields*, Sofia, Bulgaria: Union of Surveyors and Land Managers in Bulgaria (USLMB), 2004 pp. 471–479. mycite.omikk.bme.hu/doc/37231.pdf.
- [13] J. L. Awange and E. W. Grafarend, *Solving Algebraic Computational Problems in Geodesy and Geoinformatics*, Berlin: Springer, 2005.
- [14] S. Bancroft, “An Algebraic Solution of the GPS Equations,” *IEEE Transactions on Aerospace and Electronic Systems*, **AES-21**(1), 1985 pp. 56–59. dx.doi.org/doi:10.1109/TAES.1985.310538.
- [15] E. Grafarend and B. Schaffrin, *Ausgleichsrechnung in linearen Modellen*, Mannheim: B.I. Wissenschaftsverlag, 1993.
- [16] P. Singer, D. Ströbel, R. Hördt, J. Bahndorf, and K. Linkwitz, “Direkte Lösung des räumlichen Bogenschnitts,” *Zeitschrift für Vermessungswesen*, **118**, 1993 pp. 20–24.
- [17] A. Kleusberg, “Die direkte Lösung des räumlichen Hyperbelschnitts,” *Zeitschrift für Vermessungswesen*, **119**(4), 1994 pp. 188–192.
- [18] A. Kleusberg, “Analytical GPS Navigation Solution,” *Geodesy—The Challenge of the 3rd Millennium*, (E. W. Grafarend, F. W. Krumm, and V. S. Schwarze, eds.) New York: Springer, 2003 pp. 93–96.
- [19] H. Lichtenegger, “Eine direkte Lösung des räumlichen Bogenschnitts,” *Österreichische Zeitschrift für Vermessung und Geoinformation*, **83**, 1995 pp. 224–226.
- [20] J. L. Awange, “Gröbner Bases, Multipolynomial Resultants and the Gauss–Jacobi Combinatorial Algorithms—Adjustment of Nonlinear GPS/LPS Observations,” Dissertation (D93), Geodätisches Institut der Universität Stuttgart, 2002. www.ifp.uni-stuttgart.de/publications/schriftenreihe/Awange.html.
- [21] J. L. Awange and E. W. Grafarend, “Closed Form Solution of the Overdetermined Nonlinear 7 Parameter Datum Transformation,” *Allgemeine Vermessungs-Nachrichten (AVN)*, **110**, 2003 pp.130–148.
- [22] R. H. Lewis, “Heuristics to Accelerate the Dixon Resultant,” *Mathematics and Computers in Simulation*, **77**(4), 2008 pp. 400–407. dx.doi.org/doi:10.1016/j.matcom.2007.04.007.
- B. Paláncz and Z. Piroska, “A Symbolic Solution of a 3D Affine Transformation,” *The Mathematica Journal*, 2011. dx.doi.org/doi:10.3888/tmj.13-9.

About the Authors

Béla Paláncz is a professor of computer science. He received his D.Sc. from the Hungarian Academy of Sciences in 1993, and has extensive experience in education and research in the field of mathematical modeling and numeric-symbolic computations (RWTH Aachen, Germany; Imperial College, London; Wolfram Research Inc., Champaign, IL, USA).

Piroska Zaletnyik (Ph.D., Budapest University of Technology and Economics [BUTE], Hungary) is currently working as an assistant professor in the Department of Geodesy and Surveying of BUTE. She was previously a visiting scholar in Germany at Stuttgart University and at the Ohio State University. She has experience using computer algebra systems in geodesy.

Béla Paláncz

*Department of Photogrammetry and Geoinformatics
Budapest University of Technology and Economy, H - 1521, Hungary
palancz@epito.bme.hu*

Zaletnyik Piroska

*Department of Geodesy and Surveying
Budapest University of Technology and Economy, H - 1521, Hungary
zaletnyikp@gmail.com*